Programa para generar cuadraturas de Gauss − Legendre para n puntos

```
> restart: Digits:=20:
> n := 3:
> L := LegendreP(n,z): p:=expand(%);
```

$$p := \frac{5}{2} z^3 - \frac{3}{2} z \tag{1}$$

Lar raíces de los polinomios de Legendre representan las abscisas del problema

```
> x := [fsolve(p)];
```

$$x := [-0.77459666924148337704, 0., 0.77459666924148337704] \tag{2}$$

Se puede construir una procedimiento que tome como entrada una función $f(x)$ y genere los valores aproximados

```
> proced := f -> sum(c['i']*f(x['i']),'i'=1..n):
```

Se calculan los pesos a partir de resolver el siguiente sistema lineal de ecuaciones:

```
> ecs := []:
> for i from 1 to n do
>    ecs := [op(ecs),proced(z->z^(i-1)) = int(z^(i-1),z=-1..1)]:
> end do:
> ecs;
```

$$\left[c_1 + c_2 + c_3 = 2, -0.77459666924148337704\, c_1 + 0.77459666924148337704\, c_3 = 0, 0.60000000000000000001\, c_1 + 0.60000000000000000001\, c_3 = \frac{2}{3}\right]$$

```
> sys := {op(ecs)}:var := {seq(c[k],k=1..n)}:
> pesos := solve(sys,var); assign(pesos):
```

$$pesos := \{c_1 = 0.55555555555555555555, c_2 = 0.88888888888888888891, c_3 = 0.55555555555555555555\}$$

Para probar el método, generemos aleatoriamente un polinomio de grado $2\,n - 1$

```
> q := randpoly(z,degree=2*n-1);
```

$$q := -56 - 7 z^5 + 22 z^4 - 55 z^3 - 94 z^2 + 87 z \tag{3}$$

con el polinomio anterior construyamos una función $g(x)$

```
> g := unapply(q,z);
```

$$g := z \rightarrow -56 - 7 z^5 + 22 z^4 - 55 z^3 - 94 z^2 + 87 z \tag{4}$$

apliquemos la regla definida anteriormente

```
> calculado := proced(g);
```

$$calculado := -165.86666666666666667 \tag{5}$$

Calculemos el resultado exacto

```
> exacto := int(q,z=-1..1);
```

$$exacto := -\frac{2488}{15} \tag{6}$$

Ahora podemos comparar el resultado exacto con el aproximado

```
> evalf(exacto - calculado);
```

$$0. \tag{7}$$

FIN